
Example Program Listing

The following pages contain the program listing for the HP BASIC examples program. The program itself is supplied on the *HP 8510C Network Analyzer System Software Toolkit Disk*, supplied with the network analyzer. The name of the program is: EX_8510.

The program requires BASIC 5.0 or higher with the binaries IO, MAT, TRANS, and COMPLEX. The program contains many example routines, which show how to perform various programming tasks.

The disk also contains a measurement data file (BPF_DATA) that is accessed by some of the programming example routines.

Example Programs in EX_8510

1. Syntax familiarization
2. Active function output
3. Marker output
4. Marker operations
5. Single- and dual-channel displays
6. Trace data output/input
7. FORM1 data conversion
8. S11 1-port and S21 response calcs
 - 8a. Cal error coefficients
 - 8b. Modify cal set frequency subset
9. Modify cal kit
10. Simulated standard measurement
11. Using disk and tape
12. Making Plots using COPY
13. List trace values
14. Print to printer on HP 8510C system bus
15. Plot user graphics
16. Plot using BASIC HP-GL
17. Redefine parameter
18. Read and output caution/tell message
19. Read and output status bytes
20. Output key code
21. Triggered data acquisition
22. WAIT required
23. WAIT not required
24. Frequency list
25. Output/learn string
26. Input and display ASCII trace
27. Delay table operations
28. Fast CW data acquisition
29. Test port power flatness cal
30. Receiver power cal

Programming Examples

```

1 !      RE-SAVE "EX_8510"
3          ! EXAMPLES FOR INTRODUCTION TO PROGRAMMING "EX_8510"
5          !      HP8510C.07.00: January 31, 1994
7          !
9          ! Copyright © HEWLETT-PACKARD COMPANY 1984,1994
11         !      SANTA ROSA SYSTEMS DIVISION
13         !
15 OPTION BASE 0
17 DIM Formatted_data(200,1),Data(200,1)          ! 201 point trace I/O
19 DIM Data1(50,1),Data2(50,1),Data3(50,1)        ! 51 point trace I/O
21 INTEGER Form1_data(1:201,0:2)
23 INTEGER Learn_string(1:5000)
25 DIM Input$(200)
27 INTEGER Length,Error_number,Bytea,Byteb,Points,Trig,Segment
29 INTEGER Preamble,Size,Size_list,Mem
31 DIM Filename$(30),Current_line$(256),Response$(30)
33 REAL Freq,Freq2,Real,Imag,Mag,Phase,Log_mag,Lin_mag,Value
35 REAL Freq_list(400)
37 DIM Data_ascii$(200,1)[24]
39 ASSIGN @Nwa TO 716          ! Network Analyzer HP-IB Address
41 ! Read ASCII Data to/from HP 8510 HP-IB (OUTPMARK, OUTPACTI, FORM4 I/O)
43 ASSIGN @Nwa_data1 TO 716;FORMAT ON          ! (OUTPERRO, OUTPSTAT, OUTPACTI)
45 ! Read non-ASCII Data to/from HP 8510 HP-IB (FORM1, FORM2, and FORM3 I/O)
47 ASSIGN @Nwa_data2 TO 716;FORMAT OFF
49 ASSIGN @Nwa_systbus TO 717          ! Write to 8510 System Bus
51 ASSIGN @Nwa_systbusdata TO 717;FORMAT ON ! Read from HP 8510 System Bus
53 !
55 ON TIMEOUT 7,1 GOTO No_analyzer
57 ABORT 7
59 CLEAR 716
61 PRINTER IS 1
63 CONTROL KBD,2;1          ! Activate user softkeys
65 CLEAR SCREEN
67 OUTPUT @Nwa;"DEBUON; MKRLISTON; DATETIMEON; OUTPERRO;"
69 ENTER @Nwa_data1;Error_number          ! Clear Message
71 OFF TIMEOUT
73 !
75 PRINT
77 PRINT TAB(20);RPT$("*",37)
79 PRINT TAB(20);"*";TAB(56);"*"
81 PRINT TAB(20);"*  HP 8510C PROGRAMMING EXAMPLES  *"
83 PRINT TAB(20);"*";TAB(56);"*"
85 PRINT TAB(20);RPT$("*",37)
87 PRINT
89 PRINT "Note:   Refer to the HPIB Programming section of the HP 8510C Operating"
91 PRINT "       and Programming Manual for complete documentation."
93 PRINT
95 !
97 GOSUB Run_mode          ! run All or a Single example
99 !
101 LINPUT "Example 1, Input Syntax Familiarization: Press Return",Input$
103 GOSUB Example1
105 !
107 LINPUT "Example 2, Active Function Output: Press Return",Input$
109 GOSUB Example2
111 !
113 LINPUT "Example 3, Marker Output: Press Return",Input$
115 GOSUB Example3

```

```

117                                     !
119  LINPUT "Example 4, Marker Operations: Press Return",Input$
121  GOSUB Example4
123                                     !
125  LINPUT "Example 5, Single and Dual Channel Displays: Press Return",Input$
127  GOSUB Example5
129                                     !
131  LINPUT "Example 6, Trace Data Output / Input: Press Return",Input$
133  GOSUB Example6
135                                     !
137  LINPUT "Example 7, FORM1 Data Conversion: Press Return",Input$
139  GOSUB Example7
141                                     !
143  LINPUT "Example 8, S11 1-Port and S21 Response Cals: Press Return",Input$
145                                     ! Example8_a, Cal Error Coefficients
147                                     ! Example8_b, Modify Cal Set, Frequency Subset
149  GOSUB Example8
151                                     !
153  LINPUT "Example 9, Modify Cal Kit: Press Return",Input$
155  GOSUB Example9
157                                     !
159  LINPUT "Example 10, Simulated Standard Measurement: Press Return",Input$
161  GOSUB Example10
163                                     !
165  LINPUT "Example 11, Using Disc and Tape: Press Return",Input$
167  GOSUB Example11
169                                     !
171  LINPUT "Example 12, Plots Using Copy: Press Return",Input$
173  GOSUB Example12
175                                     !
177  LINPUT "Example 13, List Trace Values: Press Return",Input$
179  GOSUB Example13
181                                     !
183  LINPUT "Example 14, Print to Printer on HP 8510 System Bus: Press Return",Input$
185  GOSUB Example14
187                                     !
189  LINPUT "Example 15, Plot User Graphics: Press Return",Input$
191  GOSUB Example15
193                                     !
195  LINPUT "Example 16, Plot Using BASIC HPGL: Press Return",Input$
197  GOSUB Example16
199                                     !
201  LINPUT "Example 17, Redefine Parameter: Press Return",Input$
203  GOSUB Example17
205                                     !
207  LINPUT "Example 18, Read and Output Caution/Tell Message: Press Return",Input$
209  GOSUB Example18
211                                     !
213  LINPUT "Example 19, Read and Output Status Bytes: Press Return",Input$
215  GOSUB Example19
217                                     !
219  LINPUT "Example 20, Output Key Code: Press Return",Input$
221  GOSUB Example20
223                                     !
225  LINPUT "Example 21, Triggered Data Acquisition: Press Return",Input$
227  GOSUB Example21
229                                     !
231  LINPUT "Example 22, WAIT Required: Press Return",Input$
233  GOSUB Example22

```

Programming Examples

```

235      !
237  INPUT "Example 23, WAIT Not Required: Press Return",Input$
239  GOSUB Example23
241      !
243  INPUT "Example 24, Frequency List: Press Return",Input$
245  GOSUB Example24
247      !
249  INPUT "Example 25, Output/Input Learn String: Press Return",Input$
251  GOSUB Example25
253      !
255  INPUT "Example 26, Input and Display ASCII Trace: Press Return",Input$
257  GOSUB Example26
259      !
261  INPUT "Example 27, Delay Table Operations: Press Return",Input$
263  GOSUB Example27
265      !
267  INPUT "Example 28, FASTCW Data Acquisition: Press Return",Input$
269  GOSUB Example28
271      !
273  INPUT "Example 29, Test Port Power Flatness Cal: Press Return",Input$
275  GOSUB Example29
277      !
279  INPUT "Example 30, Receiver Power Cal: Press Return",Input$
281  GOSUB Example30
283      !
285  DISP "END OF EXAMPLES"
287  LOCAL @Hwa
289  STOP
291      !
293      ! *****
295      !
297 Example1:      ! INPUT SYNTAX FAMILIARIZATION *****
299  PRINT
301  PRINT "Example 1, Input Syntax Familiarization"
303      !
305  PRINT "  Commands"
307 Again1:      !
309  LOCAL @Hwa
311  INPUT "TYPE 8510 INSTRUCTION, THEN RETURN; ENTER 0 TO EXIT",Input$
313  IF Input$[1,1]="0" THEN
315      PRINT "  Querys"
317      GOTO Query
319  END IF
321  OUTPUT @Hwa;Input$;" "
323  PRINT Input$,
325  IF BIT(SPOLL(@Hwa),5) THEN      ! Check for syntax error
327      GOSUB Syntax_error      ! Clear error
329  END IF
331  PRINT
333  GOTO Again1

```

```

335             !
337 Query:             !
339 LOCAL @Hwa
341 LINPUT "TYPE 8510 QUERY OR OUTPUT INSTRUCTION, THEN RETURN; ENTER 0 TO EXIT ", Input$
343 IF Input$(1,1)="0" THEN
345     OUTPUT @Hwa;"OUTPERRO;"
347     ENTER @Hwa_data1;Error_number             ! Clear Message
349     RETURN
351 END IF
353 PRINT Input$,
355 OUTPUT @Hwa;Input$;";"
357 IF BIT(SPOLL(@Hwa),5) THEN                     ! Check for syntax error
359     GOSUB Syntax_error                         ! Clear error
361     PRINT
363 ELSE
365     ENTER @Hwa_data1;Input$
367     PRINT Input$
369 END IF
371 GOTO Query
373             !
375 Syntax_error:             !
377 PRINT "<< Syntax Error",
379 CLEAR @Hwa
381 OUTPUT @Hwa;"CLES; OUTPERRO;"
383 ENTER @Hwa;Error_number                     ! Clear Message
385 RETURN
387             !
389 Example2:             ! ACTIVE FUNCTION OUTPUT *****
391 PRINT
393 PRINT "Example 2, Active Function Output."
395             !
397 OUTPUT @Hwa;"USERPRES; LOGM;"
399             !
401 OUTPUT @Hwa;"STAR; OUTPACTI;"
403 ENTER @Hwa_data1;Value
405 PRINT " Start Frequency =";Value/1.E+6;" Mhz"
407             !
409 OUTPUT @Hwa;"STOP; OUTPACTI;"
411 ENTER @Hwa_data1;Value
413 PRINT " Stop Frequency =";Value/1.E+6;" Mhz"
415             !
417 OUTPUT @Hwa;"POWE; OUTPACTI;"
419 ENTER @Hwa_data1;Value
421 PRINT " Power Source 1 =";PROUND(Value,-2);" dbm"
423             !
425 OUTPUT @Hwa;"SCAL; OUTPACTI;"
427 ENTER @Hwa_data1;Value
429 PRINT "          Scale =";PROUND(Value,-2);" db/"
431             !
433 OUTPUT @Hwa;"REFV; OUTPACTI;"
435 ENTER @Hwa_data1;Value
437 PRINT " Reference Value =";PROUND(Value,-2);" db"
439             !
441 OUTPUT @Hwa;"MAGO; OUTPACTI;"
443 ENTER @Hwa_data1;Value
445 PRINT "Magnitude Offset=";Value;" db"
447             !
449 OUTPUT @Hwa;"MAGS; OUTPACTI;"
451 ENTER @Hwa_data1;Value
453 PRINT "Magnitude Slope =";Value;" db/Ghz"
455 RETURN
457

```

Programming Examples

```

459 Example3:          ! MARKER DATA OUTPUT *****
461                    !
463 PRINT
465 PRINT "Example 3, Marker Data Output"
467 DISP "Initializing System"
469 OUTPUT @Nwa;"PRES;"
471 PRINT "Averaging On, Avg Factor = 4"
473 OUTPUT @Nwa;"AVERON 4; MENUFORM; POIN 51; SING;"
475 PRINT "Automatic Holdoff For Single or Number of Groups"
477                    !
479 Again_3:           !
481 LOCAL @Nwa
483 LINPUT "Set 8510 to desired Domain, Format and Sweep Mode or E to Exit",Input$
485 IF UPC$(Input$)="E" THEN
487     OUTPUT @Nwa;"AVEROFF; FREQ; CONT;"
489     RETURN
491 END IF
493                    !
495 OUTPUT @Nwa;"SWEM?"          ! Query sweep mode
497 ENTER @Nwa_data1;Input$
499 IF Input$[2;4]="RAMP" THEN
501     OUTPUT @Nwa;"NUMG 5;"          ! NUMG = AVER factor + 1
503 ELSE
505     OUTPUT @Nwa;"SING;"          ! 8510C automatically waits until SING or NUMG
507                                ! completes before executing further instructions
509 END IF
511                    !
513 OUTPUT @Nwa;"AUTO; MARK1; MARKMAXI; OUTPMARK;"
515 ENTER @Nwa_data1;Mag,Phase          ! Read Marker Value
517 OUTPUT @Nwa;"FORM?;"          ! Query Display Format
519 ENTER @Nwa_data1;Input$
521                    !
523 PRINT " Marker ";Input$;" = ";Mag;
525 IF Phase<>0 THEN PRINT Phase;
527                    !
529 OUTPUT @Nwa;"MARK1; OUTPACTI;"
531 ENTER @Nwa_data1;Freq
533 OUTPUT @Nwa;"DOMA?;"          ! Query Domain
535 ENTER @Nwa_data1;Input$
537 SELECT Input$[2;3]
539 CASE "FRE"
541     PRINT " @";Freq/1.E+6;" Mhz"
543 CASE "AUX"
545     PRINT " @ ";Freq;" Volts"
547 CASE "TIM"
549     PRINT " @ ";Freq*1.E+9;" nano Seconds"
551 CASE "PUL"
553     PRINT " @ ";Freq*1.E+6;" micro Seconds"
555 CASE "POW"
557     PRINT " @ ";Freq;" dBm"
559 END SELECT
561 GOTO Again_3
563                    !

```

```

565 Example4:          ! MARKER OPERATIONS *****
567                   !
569   PRINT
571   PRINT "Example 4, Marker Operations"
573                   !
575   PRINT "Using = Marker"
577                   !
579   DISP "Initializing System"
581   OUTPUT @Nwa;"PRES; POIN51; SING; AUTO; CONT;"
583                   !
585   OUTPUT @Nwa;"MARK1; MARKMAXI;"
587   PRINT "Refrence Value = Marker"
589   LINPUT "Press Return for REF VALUE = MARKER",Input$
591   OUTPUT @Nwa;"REFV; EQUA;"
593                   !
595   OUTPUT @Nwa;"MARK1 5Ghz;"
597   PRINT "Start Frequency = Marker"
599   LINPUT "Press Return for START FREQ = MARKER",Input$
601   OUTPUT @Nwa;"STAR; EQUA;"
603                   !
605   OUTPUT @Nwa;"MARK2 15Ghz;"
607   PRINT "Stop Frequency = Marker"
609   LINPUT "Press Return for STOP FREQ = MARKER",Input$
611   OUTPUT @Nwa;"STOP; EQUA;"
613                   !
615   PRINT "Phase Offset = Marker"
617   LINPUT "Press Return for PHASE OFFSET = MARKER",Input$
619   OUTPUT @Nwa;"PHAS; AUTD; MARK3 10.25E9;"
621   OUTPUT @Nwa;"PHAO; EQUA;"
623                   !
625   PRINT "Peak-to-Peak Measurement."
627                   !
629   LINPUT "Press Return for Peak-to-Peak Measurement",Input$
631   OUTPUT @Nwa;"LOGM; MARKOFF; ENTO; SING; AUTO; MKRLISTON;"
633   OUTPUT @Nwa;"MARK2; MARKMAXI; DELR2; MARK1; MARKMINI; OUTPMARK;"
635   ENTER @Nwa_data1;Mag,Phase
637   OUTPUT @Nwa;"OUTPACTI;"
639   ENTER @Nwa_data1;Freq
641   PRINT "P-P Mag = ";Mag;"      P-P Freq = ";Freq
643   OUTPUT @Nwa;"MKRLISTON;"
645   LINPUT "Press Return for -3db Measurement",Input$
647   OUTPUT @Nwa;"DELO; MARKOFF; CONT;"
649                   !
651   PRINT "-3 dB Bandwidth Measurement."
653                   !
655   OUTPUT @Nwa;"S21; POIN 201; SPAN 1GHZ; CENT 10.24GHZ; SING;"
657   ASSIGN @File TO "BPF_DATA"          ! load data for band pass filter
659   ENTER @File;Preamble,Size,Data(*)
661   ASSIGN @File TO *
663   OUTPUT @Nwa;"FORM3; INPURAW1;"
665   OUTPUT @Nwa_data2;Preamble,Size,Data(*)
667                   !
669   OUTPUT @Nwa;"REFP 7; SCAL 3; MARKCONT; MARK1; MARKMAXI; REFV; EQUA;"
671   OUTPUT @Nwa;"MARK2; MARKMAXI; MARK3; MARKMAXI;"
673   OUTPUT @Nwa;"DELR1; TARV -3; MARKTARG; MARK2; SEAL;"
675   OUTPUT @Nwa;"DELR2; MARK3; OUTPACTI;"
677   ENTER @Nwa_data1;Freq
679   PRINT "-3db Bandwidth   = ";Freq/1.E+6;" MHz"

```

Programming Examples

```

681      !
683  OUTPUT @Nwa;"DELO; MARK2; OUTPACTI;"
685  ENTER @Nwa_data1;Freq
687  OUTPUT @Nwa;"MARK3; OUTPACTI;"
689  ENTER @Nwa_data1;Freq2
691  PRINT "Center Frequency = ";((Freq+Freq2)/2)/1.E+9;" GHz"
693  RETURN
695      !
697 Example5:      ! DISPLAY MODES *****
699  PRINT
701  PRINT "Example 5, Display Modes"
703      !
705      ! SINGLE CHANNEL DISPLAYS
707      !
709  PRINT "Single Channel, Four Parameter Split Display"
711  OUTPUT @Nwa;"COUC; CHAN1; FOUPSPLI; NUMG2;"
713  OUTPUT @Nwa;"S11; AUTO; S21; AUTO; S12; AUTO; S22; AUTO;"
715  OUTPUT @Nwa;"S11; CONT; MKRLFIVM; MARK1; MARK2; MARK3; MARK4; MARK5;"
717  PRINT "Marker List shows All Markers for Selected Parameter"
719      !
721  LINPUT "Press Return to Continue.",Input$
723  OUTPUT @Nwa;"MKRLFOUP; MARK3;"
725  PRINT "Marker List shows Active Marker for all Parameters"
727      !
729  LINPUT "Press Return to Continue.",Input$
731  OUTPUT @Nwa;"FOUPOVER;"
733  PRINT "Single Channel, Four Parameter Overlay Display"
735      !
737  LINPUT "Press Return for Channel 2 Display",Input$
739  OUTPUT @Nwa;"CHAN2; SING; AUTO; CONT; MARK1;"
741  PRINT "Single Channel, Single Parameter Display"
743  PRINT "Single or Four Parameter Display is Not Coupled"
745      !
747  LINPUT "Press Return for Channel 1 Display",Input$
749  OUTPUT @Nwa;"CHAN1;"
751      !
753      ! DUAL CHANNEL DISPLAYS
755      !
757  LINPUT "Press Return for Dual Channel Split Display",Input$
759  OUTPUT @Nwa;"SPLI;"
761  PRINT "Dual Channel, Single Parameter/Channel Split Display"
763      !
765  LINPUT "Press Return for Dual Channel Overlay Display",Input$
767  OUTPUT @Nwa;"OVER;"
769  PRINT "Dual Channel, Single Parameter/Channel Overlay Display"
771      !
773  LINPUT "Press Return to Continue",Input$
775      !
777      ! DUAL CHANNEL ALTERNATE SWEEP
779      !
781  PRINT "Dual Channel, Uncoupled Stimulus, Alternate Sweep"
783  OUTPUT @Nwa;"CONT; MARKOFF;"
785  OUTPUT @Nwa;"CHAN1; STAR 2 GHz; STOP 5 GHz; SING; CONT; UNCC;"
787  OUTPUT @Nwa;"CHAN2; S11; STAR 3 GHz; STOP 4 GHz;"
789  OUTPUT @Nwa;"SPLI; SING; CONT; MARK1 3.5 GHz; AUTO;"
791      !
793  LINPUT "Press Return to Continue",Input$
795  RETURN
797

```



```

!
821 Example6:          ! TRACE DATA OUTPUT / INPUT *****
823 PRINT
825 PRINT "Example 6, Trace Data Output / Input (FORM3)."
827 !
829 ! Output data from analyzer
831 !
833 OUTPUT @Hwa;"PRES;"
835 OUTPUT @Hwa;"POIN201; SPAN .5GHZ; REIP; SING; AUTO; FORM3; OUTPDATA;"
837 ENTER @Hwa_data2;Preamble,Size,Data(*)
839 !
841 PRINT "First and last data points of output corrected data array :"
843 PRINT "Point: 1";TAB(13);"Real: ";Data(0,0);TAB(36);" Imag: ";Data(0,1)
845 PRINT "Point: 201";TAB(13);"Real: ";Data(200,0);TAB(36);" Imag: ";Data(200,1)
847 !
849 OUTPUT @Hwa;"MARK1 0;"          ! set marker to first point
851 LOCAL @Hwa
853 INPUT "Corrected data array read. Press Return to Continue",Input$
855 !
857 ! Input data to analyzer
859 !
861 OUTPUT @Hwa;"ENTO; POIN201;"          ! Zero Trace for effect
863 LINPUT "Data Zeroed, Press Return To Write Data To 8510",Input$
865 !
867 OUTPUT @Hwa;"FORM3; INPUDATA;"
869 OUTPUT @Hwa_data2;Preamble,Size,Data(*)
871 PRINT "Corrected array data Written (input) to 8510."
873 RETURN
875 !
877 Example7:          ! FORM1 DATA CONVERSION *****
879 !
881 PRINT
883 PRINT "Example 7, Form1 Data Conversion"
885 !
887 ! This example reads FORM1 data (internal binary format) and converts
889 ! it to real & imaginary, linear magnitude, log magnitude and phase.
891 ! The data arrays size will automatically adjust for any number of
893 ! measurement points. Converted values are printed for the first and
895 ! last points.
897 !
899 OUTPUT @Hwa;"SING; MARK1;"
901 OUTPUT @Hwa;"FORM1; OUTPDATA;"          ! or OUTPRAWn; OUTPDATA; OUTPFORM;
903 ! or OUTPDELA; OUTPMEMO;
905 ! note: if using OUTPFORM, Data_re(I) will be in the current display
907 ! format and Data_im(I) will = 0 for all display formats that
909 ! are not plots of real / imaginary pairs. Calculated linear,
911 ! log and phase values are not valid.
913 ENTER @Hwa_data2;Preamble,Size          ! Size/6 = number of data points
915 !
917 REDIM Form1_data(1:Size/6,2)          ! dimension 0 = imag mantissa,
919 ! 1 = real mantissa and 2 = common exponent
921 !
923 ENTER @Hwa_data2;Form1_data(*)          ! read the data
925 !
927 ! Calculate Exponent - The exponent is represented by bits 0-7 of
929 ! the 16 bit integer, Form1_data(n,2). Bit 7 is the sign bit (i="-",
931 ! 0="+"). The computed value is offset by -15 to give values which
933 ! are in a useful range for measurements. Thus, for Form1_data(n,2)
935 ! values of 0 to 127, exponents range from -15 to 112 and for values
937 ! of 128 to 255, exponents range from -143 to -16 respectively. This
939 ! gives a data range of ~ 674 to -860 db using db=20*LGT(2^(exponent)).
941 ! An alternate, table method is used to decode the exponent in example28.

```

Programming Examples

```

943      !
944      FOR I=1 TO SIZE(Form1_data,1)
945          Exponent=BINAND(Form1_data(I,2),255)      ! bits 0-7 are the exponent
946          !
947          IF Exponent<128 THEN                        ! exponent is positive
948              Exponent=2^(Exponent-15)              ! offset (-15)
949              !
950          ELSE                                        ! exponent is negative
951              Exponent=2^(BINCHP(BINXOR(Exponent,255))-15) ! reverse [EOR],
952              ! change sign [CMP] and offset [-15] for negative going exponents
953          END IF
954          !
955          ! Calculate real and imaginary data
956          Real=Form1_data(I,1)*Exponent
957          Imag=Form1_data(I,0)*Exponent
958          !
959          ! Calculate linear magnitude data
960          Lin_mag=SQRT(Real^2+Imag^2)
961          !
962          ! Calculate log magnitude data
963          Log_mag=20*LGT(Lin_mag)
964          !
965          ! Calculate phase data
966          DEG
967          IF Imag=0 AND Real<0 THEN
968              Phase=-180
969          ELSE
970              Phase=2*ATN(Imag/(Real+Lin_mag))
971          END IF
972          !
973          IF I=1 OR I=SIZE(Form1_data,1) THEN      ! print first and last points
974              PRINT "Pt";I;"   Real = ";Real;"   Imag = ";Imag
975              PRINT "   Lin = ";Lin_mag;"   Log = ";Log_mag;"   Phase = ";Phase
976              PRINT
977          END IF
978      NEXT I
979      !
980      REDIM Form1_data(0:2,1:201)
981      !
982      PRINT
983      LOCAL @Hwa
984      RETURN
985      !
986      Example8:      ! S11 1-PORT AND S21 RESPONSE CALIBRATIONS *****
987      PRINT
988      PRINT "Example 8, S11 1-Port and S21 Response Calibrations"
989      !
990      OUTPUT @Hwa;"PRES; MENU CAL;"
991      INPUT "Which Cal Kit is being used (ENTER 1 or 2)?",Kit$
992      PRINT "S11 1-Port Measurement Calibration USING Cal ";Kit$
993      OUTPUT @Hwa;"CLES; CAL";Kit$;" ; CALIS111;"
994      !
995      INPUT "Port 1, Connect Shielded Open, then press Return",Input$
996      OUTPUT @Hwa;"CLASS11A;"
997      GOSUB Wait_for_meas
998      ! (Shielded Open Circuit Data Measured)

```

```

1055 LINPUT "Port 1, Connect Short, then press Return",Input$
1057 OUTPUT @Hwa;"CLASS11B;"          ! (Short Circuit Data Measured)
1059 GOSUB Wait_for_meas
1061 OUTPUT @Hwa;"CLASS11C;"          ! (Uses Both LOWBAND and SLIDING)
1063 LINPUT "Broadband OR Lowband, Slidiing Load Cal (ENTER B or S)?",Input$
1065 IF UPC$(Input$)="B" THEN
1067   LINPUT "Port1, Connect Broadband Load, then press Return",Input$
1069   OUTPUT @Hwa;"STANA;"
1071   GOSUB Wait_for_meas
1073 ELSE
1075   LINPUT "Port 1, Connect Lowband Load, then press Return",Input$
1077   OUTPUT @Hwa;"STANC;"          ! (LOWBAND Load Data Measured)
1079   GOSUB Wait_for_meas
1081   !
1083   LINPUT "Port 1, Connect Sliding Load, then press Return",Input$
1085   OUTPUT @Hwa;"STANB;"          ! (Select Sliding Load)
1087   LINPUT "Move Element to First Index Mark, then press Return",Input$
1089   OUTPUT @Hwa;"SLIS;"          ! (Sliding Load Data Measured)
1091   GOSUB Wait_for_meas
1093   FOR Slide=2 TO 6
1095     LINPUT "Move Element to Next Index Mark, then press Return",Input$
1097     OUTPUT @Hwa;"SLIS;"          ! (Sliding Load Data Measured)
1099     GOSUB Wait_for_meas
1101   NEXT Slide
1103   OUTPUT @Hwa;"SLID;"
1105 END IF
1107   !
1109 OUTPUT @Hwa;"DONE; SAV1; CALS4;"
1111   ! (Error coefficients computed and stored;
1113   !   Cal Menu displayed with CORRECTION ON;
1115   !   Corrected S11 trace displayed.)
1117 PRINT "S11 1-Port Cal Complete and Saved in Cal Set 4"
1119 LINPUT "Press Return",Input$
1121   !
1123 PRINT "S21 Response Measurement Calibration."
1125 OUTPUT @Hwa;"CHAN2; S21; CAL";Kit$;"; CALIRESP;"
1127 LINPUT "Connect Thru, then press Return",Input$
1129 OUTPUT @Hwa;"STANC;"          ! (Thru Data Measured)
1131 GOSUB Wait_for_meas
1133 OUTPUT @Hwa;"DONE; CALS5;"
1135   ! (Vector frequency response computed and stored;
1137   !   Cal Menu displayed with CORRECTION ON;
1139   !   Corrected S21 trace displayed.)
1141 PRINT "Response Cal Complete and Saved in Cal Set 5"
1143   !
1145 LINPUT "Press Return",Input$
1147   !
1149 PRINT "Corrected Measurement Device S11 and S21."
1151 LINPUT "Connect Device Under Test, then press Return",Input$
1153 OUTPUT @Hwa;"SPLI; SING; AUTO; CHAN1; AUTO; CONT;"
1155 LINPUT "Press Return",Input$
1157   !
1159 GOSUB Example8_a
1161 GOSUB Example8_b
1163   !
1165 Wait_for_meas:          ! Status Byte BIT 4 True when Standard Measured
1167 REPEAT
1169   Ser_poll=SPOLL(@Hwa)
1171   WAIT .1
1173 UNTIL BIT(Ser_poll,4)
1175 OUTPUT @Hwa;"CLES;"
1177 RETURN
1179   !

```

Programming Examples

```

1181 Example8_a:                ! CAL ERROR COEFFICIENTS *****
1183 PRINT
1185 PRINT "Example 8_a, Calibration Error Coefficients"
1187 !
1189 OUTPUT @Nwa;"PRES;"
1191 Read_response:              !
1193 PRINT "Read Cal Coefficient, Cal 5 (S21 Response Cal)"
1195 OUTPUT @Nwa;"S21; CORROW; CALS5; FORM3; OUTPCALCO1;"
1197 ENTER @Nwa_data2;Preamble,Size,Data(*)
1199 !
1201 FOR N=0 TO 200
1203     ! Data can be modified here
1205     Formatted_data(N,0)=Data(N,0)
1207     Formatted_data(N,1)=Data(N,1)
1209 NEXT N
1211 !
1213 PRINT "Write Processed Cal Coefficients."
1215 OUTPUT @Nwa;"CORROFF; CAL1; CALIRESP; FORM3; INPUCALCO1;"
1217 OUTPUT @Nwa_data2;Preamble,Size,Formatted_data(*)
1219 OUTPUT @Nwa;"SAVC; CALS6; CONT;"
1221 !
1223 PRINT " Processed Cal saved in Cal Set 6."
1225 LINPUT "Press Return",Input$
1227 !
1229 Read_1_port:              !
1231 !
1233 OUTPUT @Nwa;"DEBUOFF; HOLD; S11; CORROW; CALS4; OUTPERRO;"
1235 ENTER @Nwa;Error_number    ! clear message
1237 !
1239 PRINT "Read and Display Cal Coefficients, Cal 4 (S11 1-Port)"
1241 PRINT " Display Directivity Coefficient."
1243 OUTPUT @Nwa;"FORM3; OUTPCALCO1;"
1245 ENTER @Nwa_data2;Preamble,Size,Data(*)
1247 !
1249 OUTPUT @Nwa;"FORM3; INPUDATA;"
1251 OUTPUT @Nwa_data2;Preamble,Size,Data(*)
1253 OUTPUT @Nwa;"AUTO; TITL""PORT 1 DIRECTIVITY"";"
1255 LINPUT "Press Return",Input$
1257 !
1259 PRINT " Display Source Match Coefficient."
1261 OUTPUT @Nwa;"FORM3; OUTPCALCO2;"
1263 ENTER @Nwa_data2;Preamble,Size,Data(*)
1265 !
1267 OUTPUT @Nwa;"FORM3; INPUDATA;"
1269 OUTPUT @Nwa_data2;Preamble,Size,Data(*)
1271 OUTPUT @Nwa;"AUTO; TITL""PORT 1 SOURCE MATCH"";"
1273 LINPUT "Press Return",Input$
1275 !
1277 PRINT " Display Relection Tracking Coefficient."
1279 OUTPUT @Nwa;"FORM3; OUTPCALCO3;"
1281 ENTER @Nwa_data2;Preamble,Size,Data(*)
1283 !
1285 OUTPUT @Nwa;"FORM3; INPUDATA;"
1287 OUTPUT @Nwa_data2;Preamble,Size,Data(*)
1289 OUTPUT @Nwa;"AUTO; TITL""PORT 1 REFLECTION TRACKING"";"
1291

```

```

!
1293 LINPUT "Press Return",Input$
1295 OUTPUT @Hwa;"CONT; DEBUON;";
1297 RETURN
1299 !
1301 Example8_b: ! Modify Cal Set, Frequency Subset *****
1303 PRINT
1305 PRINT "Example 8_b, Modify Cal Set, Frequency Subset"
1307 !
1309 PRINT "Define Frequency Subset of Cal 4 (S11 1-Port)"
1311 OUTPUT @Hwa;"CORRON; CALS4;";
1313 OUTPUT @Hwa;"FRES; SUBSCENT 12 GHz; SUBSSPAN 2 GHz; CRES; CALS7;";
1315 !
1317 PRINT " Subset Defined, Cal Set 7."
1319 LINPUT "Press Return",Input$
1321 PRINT "Interactive Method to Define Frequency Subset."
1323 !
1325 OUTPUT @Hwa;"CORROFF; CORRON; CALS4; MENU CAL;";
1327 OUTPUT @Hwa;"SOFT8; SOFT4; SOFT4;"; ! freq subset menu
1329 LOCAL @Hwa
1331 LINPUT "Define Frequency Subset, then Press Return.",Input$
1333 OUTPUT @Hwa;"CRES; CALS8;";
1335 PRINT " Subset Defined, Cal Set 8."
1337 RETURN
1339 !
1341 Example9: ! MODIFY CAL KIT (TYPICAL X-BAND WAVEGUIDE) *****
1343 PRINT
1345 PRINT "Example 9, Modify Cal Kit (Typical X-Band Waveguide)"
1347 !
1349 OUTPUT @Hwa;"USERPRES; MENU CAL;";
1351 PRINT "Saving Current Cal Kit 2 to Disc."
1353 LINPUT "Insert Initialized Disc, then Return",Input$
1355 OUTPUT @Hwa;"STOIINT; STOR; CALK2; FILE2;";
1357 PRINT "Old Cal Kit 2 now on File2."
1359 !
1361 PRINT "Defining New Cal Kit 2."
1363 OUTPUT @Hwa;"MODI2; DEFS 1; STDTSOR;";
1365 OUTPUT @Hwa;"OFFD 0.018652 ns; OFFL 0; OFFZ 1;";
1367 OUTPUT @Hwa;"MINF 6.557 GHz; MAXF 999 GHz;";
1369 OUTPUT @Hwa;"WAVE; LABS""XSHORT 1""; STDD;";
1371 !
1373 OUTPUT @Hwa;"DEFS 2; STDTSOR;";
1375 OUTPUT @Hwa;"OFFD 0.055957 ns; OFFL 0; OFFZ 1;";
1377 OUTPUT @Hwa;"MINF 6.557 GHz; MAXF 999 GHz;";
1379 OUTPUT @Hwa;"WAVE; LABS""XSHORT 2""; STDD;";
1381 !
1383 OUTPUT @Hwa;"DEFS 3; STDLOAD;";
1385 OUTPUT @Hwa;"OFFD 0; OFFL 0; OFFZ 1; MINF 6.557 GHz; MAXF 999 GHz;";
1387 OUTPUT @Hwa;"FIXE; WAVE; LABS""XLOAD""; STDD;";
1389 !
1391 OUTPUT @Hwa;"DEFS 11; STDDELA;";
1393 OUTPUT @Hwa;"OFFD 0; OFFL 0; OFFZ 1; MINF 6.557 GHz; MAXF 999 GHz;";
1395 OUTPUT @Hwa;"WAVE; LABS""XTHRU""; STDD;";
1397 !
1399 OUTPUT @Hwa;"SPECS11A 1; CLAD; SPECS11B 2; CLAD; SPECS11C 3; CLAD;";
1401 !
1403 OUTPUT @Hwa;"LABES11A""1/8 SHORT"";";
1405 OUTPUT @Hwa;"LABES11B""3/8 SHORT"";";
1407 OUTPUT @Hwa;"LABES11C""STDILD"";";
1409

```

Programming Examples

```

!
1411 OUTPUT @Hwa;"SPECS22A 1; CLAD; SPECS22B 2; CLAD; SPECS22C 3; CLAD;"
1413      !
1415 OUTPUT @Hwa;"LABES22A""1/8 SHORT"";"
1417 OUTPUT @Hwa;"LABES22B""3/8 SHORT"";"
1419 OUTPUT @Hwa;"LABES22C""STDOLD"";"
1421      !
1423 OUTPUT @Hwa;"SPECFWDT 11; CLAD; SPECREVT 11; CLAD;"
1425 OUTPUT @Hwa;"SPECFWD 11; CLAD; SPECREVM 11; CLAD;"
1427 OUTPUT @Hwa;"SPECRESP 1, 2, 11; CLAD;"
1429      !
1431 OUTPUT @Hwa;"LABEFWD"" THRU ""; LABEREVT"" THRU "";"
1433 OUTPUT @Hwa;"LABEFWD"" THRU ""; LABEREVM"" THRU "";"
1435 OUTPUT @Hwa;"LABERESP""RESPONSE"";"
1437      !
1439 OUTPUT @Hwa;"LABK"" X BAND "";"
1441 OUTPUT @Hwa;"KITD;"
1443      !
1445 OUTPUT @Hwa;"STOR; CALK2; FILE3; MENUCL;"
1447 PRINT "New Cal Kit now in File3."
1449      !
1451 LINPUT "Reload Old cal kit? (Y/N)",Input$
1453 IF UPC$(Input$)="N" THEN RETURN
1455 OUTPUT @Hwa;"LOAD; CALK2; FILE2; MENUCL;"
1457 PRINT "Prior Cal Kit 2 Reloaded."
1459      !
1461 RETURN
1463      !
1465 Example10:      ! SIMS CALIBRATION *****
1467 PRINT
1469 PRINT "Example 10, Simulated Calibration"
1471      !
1473 PRINT "Measure Standards Data Used For Simulated Cal."
1475 OUTPUT @Hwa;"PRES; STEP; POINS1; AVERON 4; ENTO;"
1477      !
1479 LINPUT "Connect Short to Port 1, then Return",Input$
1481 OUTPUT @Hwa;"SING; FORM3; OUTPDATA;"
1483 ENTER @Hwa_data2;Preamble,Size,Data1(*)      ! std 1 data
1485      !
1487 LINPUT "Connect Open to Port 1, then Return",Input$
1489 OUTPUT @Hwa;"SING; FORM3; OUTPRAW1;"
1491 ENTER @Hwa_data2;Preamble,Size,Data2(*)      ! std 2 data
1493      !
1495 LINPUT "Connect Broadband Load to Port 1, then Return",Input$
1497 OUTPUT @Hwa;"SING; FORM3; OUTPRAW1;"
1499 ENTER @Hwa_data2;Preamble,Size,Data3(*)      ! std 3 data
1501 OUTPUT @Hwa;"CONT;"
1503      !
1505      ! Begin SIMS S11 1-Port Calibration
1507      !
1509 OUTPUT @Hwa;"CLES; TRIG;"      ! Set Up FOR SIMS;
1511 OUTPUT @Hwa;"CAL1; CALIS111"      ! Select Cal Type
1513 OUTPUT @Hwa;"CLASS11A;"      ! Select Standard (short)
1515 GOSUB Wait_for_trig      ! Wait for Bit 2 then CLEAR 716
1517 CLEAR @Hwa
1519 OUTPUT @Hwa;"FORM3; INPURAW1;"      ! Input simulated Standard data
1521 OUTPUT @Hwa_data2;Preamble,Size,Data1(*)
1523 OUTPUT @Hwa;"SIMS;"      ! Input Complete
1525      !
1527 OUTPUT @Hwa;"CLASS11B;"      ! Select Next Standard (open)
1529 GOSUB Wait_for_trig
1531 CLEAR @Hwa
1533 OUTPUT @Hwa;"FORM3; INPURAW1;"
1535 OUTPUT @Hwa_data2;Preamble,Size,Data2(*)
1537 OUTPUT @Hwa;"SIMS;"
1539

```

```

!
1541 OUTPUT @Hwa;"CLASS11C; STANA;"      ! Select Next Standard (broadband)
1543 GOSUB Wait_for_trig
1545 CLEAR @Hwa
1547 OUTPUT @Hwa;"FORM3; INPURAW1;"
1549 OUTPUT @Hwa_data2;Preamble,Size,Data3(*)
1551 OUTPUT @Hwa;"SIMS;"
1553 !
1555 OUTPUT @Hwa;"DONE;"                  ! Standard Class Complete
1557 OUTPUT @Hwa;"SAV1; CALS3;"          ! Cal Complete and Saved
1559 OUTPUT @Hwa;"FRER;"                  ! Return to normal sweeping
1561 !
1563 PRINT "Simulated Cal Complete and Saved in Cal Set 3"
1565 RETURN
1567 !
1569 Wait_for_trig:      ! Bit 2, ready for trigger
1571 REPEAT
1573   Ser_poll=SPOLL(@Hwa)
1575 UNTIL BIT(Ser_poll,2)
1577 RETURN
1579 !
1581 Example11:          ! USING DISC / READ CITIFILE *****
1583 PRINT
1585 PRINT "Example 11, Using Disc / Read CITIfile"
1587 OUTPUT @Hwa;"PRES; POIN51; USER1; SING; DATI; S11; SPLI; SING;"
1589 !
1591 Disc:                !
1593 LINPUT "Internal or External Disc? (I or E).",Input$
1595 IF UPC$(Input$)="E" THEN
1597   OUTPUT @Hwa;"STOIEXT;"              ! Use External Disc *****
1599   LINPUT "Insert Disc in External Drive, then Return",Input$
1601 ELSE
1603   OUTPUT @Hwa;"STOIINT;"              ! Use Internal Disc *****
1605   LINPUT "Insert Disc in Internal Drive, then Return",Input$
1607 END IF
1609 OUTPUT @Hwa;"SAVE1;"                  ! save set-up in instrument state 1
1611 !
1613 Initdisc:            !
1615 LINPUT "Initialize Disc? (ENTER Y or N)",Input$
1617 IF UPC$(Input$)="Y" THEN OUTPUT @Hwa;"INID;"
1619 !
1621 Storedisc:          !
1623 PRINT "Store Data to Disc"
1625 OUTPUT @Hwa;"STOR; INSS1; DISF ""IFILE1"";"
1627 OUTPUT @Hwa;"CHAN1; STOR; DATARAW; DISF ""DFILE1"";"
1629 OUTPUT @Hwa;"CHAN2; STOR; DATARAW; DISF ""DFILE2"";"
1631 OUTPUT @Hwa;"STOR; MEMO1; DISF ""MFILE1"";"
1633 OUTPUT @Hwa;"STOR; CALK1; DISF ""KFILE1"";"
1635 OUTPUT @Hwa;"DIRE;"
1637 LOCAL @Hwa
1639 LINPUT "Directory Displayed, Press Return to Load Data",Input$
1641

```

Programming Examples

```

!
1643 Loaddisc:
1645 PRINT "Load Data From Disc"
1647 OUTPUT @Nwa;"LOAD; INSS1; DISF ""IFILE1"";";
1649 OUTPUT @Nwa;"RECA1;";
1651 PRINT "HOLD Avoids Overwritting Data Loaded From Disc."
1653 LINPUT "Press Return",Input$
1655 OUTPUT @Nwa;"CHAN1; LOAD; DATARAW; DISF ""DFILE1"";";
1657 OUTPUT @Nwa;"CHAN2; LOAD; DATARAW; DISF ""DFILE2"";";
1659 OUTPUT @Nwa;"CHAN2; DISPDAT; CHAN1; DISPDAT;";
1661 PRINT "Must Turn Both Channel's Memories Off Before Loading any Memory."
1663 OUTPUT @Nwa;"LOAD; MEMO1; DISF ""MFILE1"";";
1665 OUTPUT @Nwa;"DISPDATM;";
1667 !
1669 LINPUT "Print Contents of a CITIfile? (ENTER Y or N) External Drive Required on Controller Bus",Input$
1671 IF UPC$(Input$)<>"Y" THEN RETURN
1673 OUTPUT @Nwa;"DIRE;";
1675 LINPUT "Output to Printer or Controller CRT? (ENTER P or C)",Input$
1677 IF UPC$(Input$)="P" THEN
1679     LINPUT "Is Printer on 8510 System Bus? (ENTER Y or N)",Input$
1681     IF UPC$(Input$)="Y" THEN
1683         PRINTER IS 717
1685         OUTPUT @Nwa;"ADDRPASS 01;";
1687     ELSE
1689         PRINTER IS 701                ! Connected to Controller HP-IB
1691     END IF
1693 ELSE
1695     PRINTER IS 1                    ! Print to Controller CRT
1697 END IF
1699 !
1701 LINPUT "INSTALL DISC IN CONTROLLER DRIVE 0, THEN RETURN.",Input$
1703 Citiread:
1705 LINPUT "NAME OF CITIfile to Read?",File_name$
1707 ON ERROR GOSUB File_error
1709 ASSIGN @Discfile TO File_name$&":,700,0"
1711 ON END @Discfile GOTO End_of_file
1713 PRINT "DISC FILE NAME=",File_name$
1715 PRINT
1717 LOOP
1719     ENTER @Discfile;Current_line$
1721     PRINT Current_line$
1723 END LOOP
1725 End_of_file:
1727 PRINT
1729 PRINT "END OF FILE"
1731 PRINTER IS 1
1733 LINPUT "Print Another CITIfile? (ENTER Y or N)",Input$
1735 IF UPC$(Input$)="Y" THEN GOTO Citiread
1737 OFF ERROR
1739 RETURN
1741 !
1743 File_error:
1745 IF ERR=56 OR ERR=53 OR ERR=58 THEN    ! file undefined or wrong type
1747     IF ERR=56 OR ERR=53 THEN
1749         PRINT "File ";File_name$;" Not Found. Check Directory On 8510 Display"
1751     ELSE
1753         ! ERR=58
1755         PRINT "File TYPE Must Be ASC. Check Directory On 8510 Display."
1757     END IF
1759     BEEP 300,.1
1761     LINPUT "NAME OF CITIfile to Read?",File_name$
1763 ELSE
1765     OFF ERROR
1767 END IF
1769 RETURN

```



```

!
1771 Example12:      ! PLOTS USING COPY *****
1773 PRINT
1775 PRINT "Example 12, Plots Using Copy"
1777 PRINT "Requires Properly Addressed 8510C Plotter"
1779 INPUT "Skip This Example ? (ENTER Y or N)",Input$
1781 IF UPC$(Input$)<>"N" THEN RETURN
1783      !
1785 OUTPUT @Hwa;"DEBUOFF; PRES; POIN51; MARK1; FOUPSPLI; ENTO; OUTPERRO;"
1787 ENTER @Hwa;Error_number      ! clear message
1789 INPUT "Load Paper, then Return",Input$
1791 PRINT "Press any HP 8510C key to ABORT Plot."
1793 OUTPUT @Hwa;"SING; AUTO; S21; AUTO; S12; AUTO; S22; AUTO;"
1795 OUTPUT @Hwa;"FULP; PLOTALL;"
1797      !
1799 INPUT "Wait until Plotter has finished Plotting, then Press Return",Input$
1801 RETURN
1803      !
1805 Example13:      ! TRACE LIST TO PRINTER *****
1807 PRINT
1809 PRINT "Example 13, List Trace Values to System Printer."
1811      !
1813 PRINT "Requires Properly Addressed 8510C Printer"
1815 INPUT "Skip This Example ? (ENTER Y or N)",Input$
1817 IF UPC$(Input$)="Y" THEN RETURN
1819      !
1821 PRINT "Printing 51 Point Trace, with Skip Factor of 7"
1823 OUTPUT @Hwa;"SINC; S21; LIMP; POIN51; SING; AUTO;"
1825 OUTPUT @Hwa;"LISSKIP 7; LIST;"
1827 RETURN
1829      !
1831 Example14:      ! PRINT TO PRINTER ON 8510 SYSTEM BUS ***
1833 PRINT
1835 PRINT "Example 14, Print / Plot To 8510C System Bus"
1837 PRINT "Requires Printer and Plotter on HP-IB System Bus"
1839 INPUT "Skip This Example ? (ENTER Y or N)",Input$
1841 IF UPC$(Input$)<>"N" THEN RETURN
1843      !
1845 PRINT "Print Title via Pass-Thru."
1847 OUTPUT @Hwa;"ADDRPASS 01;"
1849 PRINTER IS 717      ! (Hwa_systbus)
1851 PRINT
1853 PRINT "MEASUREMENT NUMBER 1"
1855 PRINT
1857 OUTPUT @Hwa;"ENTO;"
1859 PRINTER IS 1
1861 INPUT "Press Return",Input$
1863      !
1865 Example14a:      ! PLOT TO PLOTTER ON 8510 SYSTEM BUS *****
1867      !
1869 PRINT "Plot Label via Pass-Thru."
1871 OUTPUT @Hwa;"ADDRPASS 05;"
1873 OUTPUT @Hwa_systbus;"CS;PU;PA 2500,2500;PD;LB PASS-THRU;PU;"
1875 OUTPUT @Hwa;"ENTO;"
1877 INPUT "Press Return",Input$
1879 RETURN
1881

```

Programming Examples

```
!
1883 Example15:      ! PLOT USER DISPLAY USING HP-GL SUBSET (8510C) *****
1885      !
1887      !              8510C:      X=0-5733      Y=0-4095
1889      !
1891 Plot_absolute:      !
1893 PRINT
1895 PRINT "Example 15, User Display."
1897 OUTPUT @Hwa;"ADDRPASS 31;"
1899 OUTPUT @Hwa_systbus;"PG; CS; DF;"      ! User display on and clear
1901      !
1903 OUTPUT @Hwa_systbus;"SP1; PA 0,0; PD;"
1905 OUTPUT @Hwa_systbus;"PA 0,4095, 5733,4095, 5733,0, 0,0;"
1907 OUTPUT @Hwa_systbus;"PU; PA 2475,3950; PD; LBFULL SCREENA;"
1909 OUTPUT @Hwa_systbus;"PU;"
1911 LINPUT "Press Return",Input$
1913      !
1915 OUTPUT @Hwa_systbus;"SP2; PA 180,384; PD;"
1917 OUTPUT @Hwa_systbus;"PA 180,3585, 4660,3585, 4660,384, 180,384;"
1919 OUTPUT @Hwa_systbus;"PU; PA 2420,1980; PD;"      ! Polar Center
1921 GOSUB Draw_cross
1923 OUTPUT @Hwa_systbus;"PU; PA 2000,3300; PD; LBSINGLE CHANNELA;"
1925 OUTPUT @Hwa_systbus;"PU;"
1927 LINPUT "Press Return",Input$
1929      !
1931 OUTPUT @Hwa_systbus;"SP3; PA 180,1180; PD;"
1933 OUTPUT @Hwa_systbus;"PA 180,2780, 2420,2780, 2420,1180, 180,1180;"
1935 OUTPUT @Hwa_systbus;"PU; PA 1300,1980; PD;"      ! Polar Center
1937 GOSUB Draw_cross
1939 OUTPUT @Hwa_systbus;"PU; PA 250,1500; PD; LBDUAL, CHANNEL 1A;"
1941 OUTPUT @Hwa_systbus;"PU;"
1943 LINPUT "Press Return",Input$
1945      !
1947 OUTPUT @Hwa_systbus;"SP4; PA 2465,1180; PD;"
1949 OUTPUT @Hwa_systbus;"PA 2465,2780, 4705,2780, 4705,1180, 2465,1180;"
1951 OUTPUT @Hwa_systbus;"PU; PA 3585,1980; PD;"      ! Polar Center
1953 GOSUB Draw_cross
1955 OUTPUT @Hwa_systbus;"PU; PA 2665,1500; PD; LBDUAL, CHANNEL 2A;"
1957 OUTPUT @Hwa_systbus;"PU;"
1959 LINPUT "Press Return",Input$
1961      !
1963 OUTPUT @Hwa_systbus;"SP5; PA 180,210; PD;"
1965 OUTPUT @Hwa_systbus;"PA 180,1760, 2335,1760, 2335,210, 180,210;"
1967 OUTPUT @Hwa_systbus;"PU; PA 1255,980; PD;"      ! Polar Center
1969 GOSUB Draw_cross
1971 OUTPUT @Hwa_systbus;"SP6; PU; PA 180,2260; PD;"
1973 OUTPUT @Hwa_systbus;"PA 180,3805, 2335,3805, 2335,2260, 180,2260;"
1975 OUTPUT @Hwa_systbus;"PU; PA 1255,3030; PD;"      ! Polar Center
1977 GOSUB Draw_cross
1979 OUTPUT @Hwa_systbus;"SP7; PU; PA 2510,2260; PD;"
1981 OUTPUT @Hwa_systbus;"PA 2510,3805, 4665,3805, 4665,2260, 2510,2260;"
1983 OUTPUT @Hwa_systbus;"PU; PA 3590,3030; PD;"      ! Polar Center
1985 GOSUB Draw_cross
1987 OUTPUT @Hwa_systbus;"SP9; PU; PA 2510,210; PD;"
1989 OUTPUT @Hwa_systbus;"PA 2510,1760, 4665,1760, 4665,210, 2510,210;"
1991 OUTPUT @Hwa_systbus;"PU; PA 3590,980; PD;"      ! Polar Center
1993 GOSUB Draw_cross
1995 OUTPUT @Hwa_systbus;"SP5; PU; PA 250,500; PD; LBFOUR PARAMETERA;"
1997 OUTPUT @Hwa_systbus;"PU;"
1999 LINPUT "Press Return",Input$
2001
```

```

!
2003 OUTPUT @Hwa_systbus;"SP10; PA 4870,0; PD;"
2005 OUTPUT @Hwa_systbus;"PA 4870,4095, 5733,4095, 5733,0, 4870,0;"
2007 OUTPUT @Hwa_systbus;"PU; PA 4930,2000; PD; LBMENU AREAA;"
2009 OUTPUT @Hwa_systbus;"PU;"
2011 !
2013 LINPUT "Press Return For Measurement Display On",Input$
2015 !
2017 OUTPUT @Hwa;"SINC; MENUDDOMA; ENT0;"
2019 OUTPUT @Hwa_systbus;"RS;" ! measurement display on
2021 !
2023 LINPUT "Insert Initialized Disc in 8510C Drive: Press Return",Input$
2025 PRINT "Store User Display to Disc."
2027 OUTPUT @Hwa;"STOIINT; STOR; USED; FILE1;"
2029 !
2031 LINPUT "Press Return For Measurement Display Off",Input$
2033 OUTPUT @Hwa_systbus;"CS;" ! measurement display off
2035 !
2037 LINPUT "Erase User Display: Press Return",Input$
2039 OUTPUT @Hwa_systbus;"PG;"
2041 !
2043 LINPUT "Press Return For Measurement Display On",Input$
2045 OUTPUT @Hwa_systbus;"RS;"
2047 !
2049 LINPUT "Load User Display from Disc: Press Return",Input$
2051 PRINT "Load User Display from Disc."
2053 OUTPUT @Hwa;"STOIINT; LOAD; USED; FILE1;"
2055 !
2057 LINPUT "Next Example: Press Return",Input$
2059 OUTPUT @Hwa_systbus;"PG; RS;"
2061 RETURN
2063 !
2065 Draw_cross: !
2067 OUTPUT @Hwa_systbus;"PR -200,0, 400,0, -200,0;"
2069 OUTPUT @Hwa_systbus;"PR 0,-200, 0,400, 0,-200;"
2071 RETURN
2073 !
2075 Example16: ! PLOT TO USER DISPLAY USING BASIC HP-GL *****
2077 PRINT
2079 PRINT "Example 16, Plot Using BASIC HP-GL"
2081 !
2083 OUTPUT @Hwa;"SINC; FULP; ADDRPASS 31;"
2085 OUTPUT @Hwa_systbus;"PG; CS;"
2087 PLOTTER IS 717,"HPGL"
2089 WINDOW 0,4095,0,4095
2091 !
2093 ! HP-GL PLOTTING STATEMENTS
2095 !
2097 FRAME
2099 MOVE 100,100
2101 DRAW 3995,3995
2103 MOVE 3995,100
2105 DRAW 100,3995
2107 MOVE 1600,800
2109 LABEL "BASIC HP-GL"
2111 !
2113 LINPUT "Output Display To 8510C Plotter ? (ENTER Y or N)",Input$
2115 IF UPC$(Input$)="Y" THEN
2117 OUTPUT @Hwa;"FULP; PLOTALL;"
2119 END IF
2121 !

```

Programming Examples

```
2123 OUTPUT @Hwa_systbus;"PG; RS;"
2125 RETURN
2127 !
2129 Example17: ! REDEFINE PARAMETER *****
2131 PRINT
2133 PRINT "Example 17, Redefine Parameter"
2135 !
2137 OUTPUT @Hwa;"PRES; POIN51; FOUPSPLI; DEBUOFF; ENTO;"
2139 OUTPUT @Hwa;"USER1; USER2; LOCKA2; DRIVPORT2; REDD;"
2141 OUTPUT @Hwa;"USER3; LOCKA2; DRIVPORT2; REDD; USER4;"
2143 OUTPUT @Hwa;"SAVE5;"
2145 PRINT "Redefined USER Parameters now Saved in INST STATE 5."
2147 !
2149 ! PRESET selects standard User parameter definition.
2151 ! RECALL 5 selects saved user parameter definitions.
2153 !
2155 RETURN
2157 !
2159 Example18: ! READ AND OUTPUT CAUTION/TELL MESSAGE *****
2161 PRINT
2163 PRINT "Example 18, Read and Output Caution/Tell Message"
2165 !
2167 LOOP
2169 LOCAL @Hwa
2171 LINPUT "Adjust 8510C & Press Return to Read Caution/Tell (E to Exit)",Input$
2173 EXIT IF UPC$(Input$)="E"
2175 OUTPUT @Hwa;"OUTPERR0;"
2177 ENTER @Hwa_data1;Error_number,Input$
2179 PRINT Error_number,Input$
2181 END LOOP
2183 RETURN
2185 !
2187 Example19: ! READ AND OUTPUT STATUS BYTES *****
2189 PRINT
2191 PRINT "Example 19, Read and Output Status Bytes"
2193 LOOP
2195 OUTPUT @Hwa;"cles;"
2197 LOCAL @Hwa
2199 LINPUT "Adjust 8510C & Press Return to Read Status (E to Exit)",Input$
2201 EXIT IF UPC$(Input$)="E"
2203 OUTPUT @Hwa;"OUTPSTAT;" ! output and clear status
2205 ENTER @Hwa_data1;Bytea,Byteb
2207 PRINT "Primary =" ;Bytea,"Extended =" ;Byteb
2209 END LOOP
2211 RETURN
2213 !
2215 Example20: ! OUTPUT KEY CODE *****
2217 PRINT
2219 PRINT "Example 20, Output Key Code"
2221 !
2223 OUTPUT @Hwa;"DEBUON; CLES; SRQM 128,2;" ! set mask for key press
2225 ON INTR 7 GOSUB Key_code
2227 ENABLE INTR 7;2
2229 GOSUB Blank_keys
2231 ON KEY 5 LABEL " NEXT EXAMPLE" GOTO Exit_example20
2233 DISP "PRESS HP 8510 Front Panel Key. (f5 to EXIT.)"
2235 GOTO Wait_loop
2237
```

```

!
2239 Exit_example20:  !
2241 DISABLE INTR 7
2243 GOSUB Keys_off
2245 PRINT ""
2247 RETURN
2249  !
2251 Key_code:      !
2253 Ser_poll=SPOLL(@Hwa)
2255 OUTPUT @Hwa;"OUTPKY;"
2257 ENTER @Hwa_data1;A
2259 PRINT A;
2261 ENABLE INTR 7
2263 RETURN
2265  !
2267 Example21:  ! TRIG Mode, TRIGGERED DATA ACQUISITION *****
2269 PRINT
2271 PRINT "Example 21, TRIG Mode, Triggered Data Acquisition"
2273  !
2275 DISP "Initializing System"
2277 OUTPUT @Hwa;"PRES;"
2279 Start_21:  !
2281 LINPUT "Press Return to start Triggered sweep.",Input$
2283  !
2285 OUTPUT @Hwa;"CLES; SRQM 4,0;"  ! ready for trigger bit
2287 OUTPUT @Hwa;"POIN; OUTPACTI;"
2289 ENTER @Hwa_data1;Points  ! number of points in sweep
2291 OUTPUT @Hwa;"ENTO;"
2293 ON INTR 7 GOTO Next_point
2295 ENABLE INTR 7;2
2297 Trig=0
2299 OUTPUT @Hwa;"STEP; TRIG;"  ! triggered step sweep mode
2301 GOTO Wait_loop
2303  !
2305 Next_point:  !
2307 Ser_poll=SPOLL(716)
2309 IF Trig=Points THEN GOTO End_of_sweep
2311 TRIGGER 716  ! measure a point
2313 Trig=Trig+1
2315 DISP "Trigger =";Trig
2317 ENABLE INTR 7
2319 GOTO Wait_loop
2321  !
2323 End_of_sweep:  !
2325 OFF INTR 7
2327 PRINT "End Of Sweep,";Trig;" points measured"
2329 LINPUT "Another Sweep? (ENTER Y or N)",Input$
2331 IF UPC$(Input$)="Y" THEN GOTO Start_21
2333 DISABLE INTR 7
2335 OUTPUT @Hwa;"FRER;"
2337 OUTPUT @Hwa;"CLES; SRQM 0,0"
2339 RETURN
2341

```

Programming Examples

```

!
2343 Example22:  ! WAIT Required *****
2345 PRINT
2347 PRINT "Example 22, WAIT Required for display updates."
2349 !
2351 GOSUB Blank_keys
2353 ON KEY 5 LABEL " NEXT EXAMPLE" GOTO Exit_example22
2355 !
2357 OUTPUT @Nwa;"DEBUOFF; FOUPOVER; STEP; POIN101; SING;"
2359 OUTPUT @Nwa;"S11; LIMP; DATI; DISPMATH; PHAO 0;"
2361 OUTPUT @Nwa;"S21; LIMP; DATI; DISPMATH; PHAO 90;"
2363 OUTPUT @Nwa;"S12; LIMP; DATI; DISPMATH; PHAO 180;"
2365 OUTPUT @Nwa;"S22; LIMP; DATI; DISPMATH; PHAO 270;"
2367 OUTPUT @Nwa;"OUTPERRO;"
2369 ENTER @Nwa;Error_number  ! clear message
2371 !
2373 T=0                      ! initial tint increment value
2375 M=2.5E-11                ! electrical delay increment
2377 !
2379 Eled:  !
2381 FOR N=0 TO 1 STEP M
2383   FOR P=1 TO 4
2385     SELECT P              ! Choose Parameter
2387     CASE 1
2389       OUTPUT @Nwa;"S11; COLRS11D;"
2391       T1=T+0
2393     CASE 2
2395       OUTPUT @Nwa;"S21; COLRS21D;"
2397       T1=T+25
2399     CASE 3
2401       OUTPUT @Nwa;"S12; COLRS12D;"
2403       T1=T+50
2405     CASE 4
2407       OUTPUT @Nwa;"S22; COLRS22D;"
2409       T1=T+75
2411     END SELECT
2413     IF T1>100 THEN T1=T1-100
2415     OUTPUT @Nwa;"TINT";INT(T1);"      ! Change Color
2417     OUTPUT @Nwa;"ELED";P+N;"s;"      ! Increment Delay
2419     !
2421     OUTPUT @Nwa;"WAIT; ENT0;"  ! This WAIT insures that the 8510C updates
2423                               ! the display before executing more commands
2425     !
2427     T=T+.25  ! tint value increment
2429     IF T>100 THEN T=0
2431   NEXT P
2433 NEXT N
2435 GOTO Eled
2437 !
2439 Exit_example22:  !
2441 OUTPUT @Nwa;"DEFC;"      ! Default Colors
2443 GOSUB Keys_off
2445 RETURN
2447 !
2449 Example23:  ! WAIT Not Required (holdoff included in OUTPxxxx) *****
2451 PRINT
2453 PRINT "Example 23, Wait Not Required (OUTPUxxxx holds off further execution)"
2455 !
2457 GOSUB Blank_keys
2459 ON KEY 5 LABEL " NEXT EXAMPLE" GOTO Exit_example23
2461 !

```

```

2463 OUTPUT @Hwa;"USERPRES; LOGM; SING; AUTO; STAR; OUTPACTI;"
2465 ENTER @Hwa;Freq1
2467 OUTPUT @Hwa;"STOP; OUTPACTI;"
2469 ENTER @Hwa;Freq2
2471 !
2473 OUTPUT @Hwa;"MARK1"
2475 LOOP
2477   FOR N=Freq1 TO Freq2 STEP (Freq2-Freq1)/50
2479     OUTPUT @Hwa;N;" " ! move marker to new frequency
2481     OUTPUT @Hwa;"OUTPACTI;"
2483     ENTER @Hwa;Freq ! read current marker frequency
2485     OUTPUT @Hwa;"OUTPMARK;"
2487     ENTER @Hwa;Mag,Phase
2489     DISP "Marker is ";Mag,Phase;" @ ";Freq;"Ghz"
2491   NEXT N
2493 END LOOP
2495 !
2497 Exit_example23: !
2499 GOSUB Keys_off
2501 OUTPUT @Hwa;"CONT;"
2503 RETURN
2505 !
2507 Example24: ! FREQUENCY LIST *****
2509 PRINT
2511 PRINT "Example 24, Frequency List"
2513 !
2515 OUTPUT @Hwa;"S21;"
2517 OUTPUT @Hwa;"EDITLIST; CLEL;"
2519 OUTPUT @Hwa;"SADD; STAR 2 GHz; STOP 4 GHz; STPSIZE 100 MHz; SDON;"
2521 OUTPUT @Hwa;"SADD; STAR 4 GHz; STOP 8 GHz; STPSIZE 200 MHz; SDON;"
2523 OUTPUT @Hwa;"SADD; STAR 8 GHz; STOP 16 GHz; STPSIZE 400 MHz; SDON;"
2525 OUTPUT @Hwa;"DUPD; EDITDONE;"
2527 OUTPUT @Hwa;"LISFREQ; SING; AUTO; CONT; SAVE7;"
2529 !
2531 PRINT "Frequency List set-up Complete and Saved in Instrument State 7."
2533 LINPUT "Press Return to Read Frequency List and Data.",Input$
2535 !
2537 OUTPUT @Hwa;"POIN; OUTPACTI;"
2539 ENTER @Hwa_data1;Points
2541 REDIM Freq_list(Points-1),Data(Points-1,1)
2543 !
2545 PRINT "Read Frequency List and Data from HP 8510C."
2547 OUTPUT @Hwa;"FORM3; OUTPFREL;"
2549 ENTER @Hwa_data2;Preamble,Size_list,Freq_list(*)
2551 OUTPUT @Hwa;"FORM3; OUTPDATA;"
2553 ENTER @Hwa_data2;Preamble,Size,Data(*)
2555 !

```

Programming Examples

```
2557 PRINT "Selected Unformatted Data from";Points;" Point Frequency List"
2559 FOR I=0 TO Points-1 STEP INT(Points/2)
2561   PRINT "Point";I+1;" is ";Data(I,0);Data(I,1);" @ ";Freq_list(I)
2563 NEXT I
2565 !
2567 LOOP
2569   INPUT "Enter Segment to Sweep (1-3) (0 to Exit).",Segment
2571 EXIT IF Segment=0
2573   OUTPUT @Hwa;"CONT; SSEG";Segment;";"
2575   OUTPUT @Hwa;"SEG?;"
2577   ENTER @Hwa_data1;Input$
2579   OUTPUT @Hwa;"SEGM; OUTPACTI;"
2581   ENTER @Hwa_data1;Segment
2583   PRINT "Sweeping ";Input$;Segment
2585 END LOOP
2587 !
2589 REDIM Data(200,1)
2591 RETURN
2593 !
2595 Example25:      ! Learn String *****
2597 PRINT
2599 PRINT "Example 25, Learn String"
2601 !
2603 DISP "Initializing System"
2605 OUTPUT @Hwa;"PRES;"
2607 LOCAL @Hwa
2609 LINPUT "Set State to Save then Press Return.",Input$
2611 OUTPUT @Hwa;"OUTPLEAS;"          ! Always FORM1
2613 ENTER @Hwa_data2;Preamble,Size
2615 PRINT "Learn String Length=";Size;"Bytes"
2617 REDIM Learn_string(1:Size/2)      ! Size Depends Upon Firmware Version
2619 ENTER @Hwa_data2;Learn_string(*)
2621 OUTPUT @Hwa;"PRES;"
2623 !
2625 LINPUT "Press Return to Recall Previous Instrument State.",Input$
2627 OUTPUT @Hwa;"INPULEAS;"
2629 OUTPUT @Hwa_data2;Preamble,Size,Learn_string(*)
2631 RETURN
2633 !
2635 Example26:      ! Input Floating Point or ASCII Data *****
2637 PRINT
2639 PRINT "Example 26, Input Floating Point or ASCII Data"
2641 !
2643 GOSUB Blank_keys
2645 ON KEY 5 LABEL " NEXT EXAMPLE" GOSUB Finish
2647 !
2649 OUTPUT @Hwa;"HOLD; POIN201; SINC; S11; LINP; ENT0;"
2651 !
2653 OUTPUT 716;"FORM3; OUTPDATA;"      ! Get Preamble and Size for Form 3 Input
2655 ENTER @Hwa_data2;Preamble,Size
2657 OUTPUT @Hwa;"ENT0;"
2659 !
2661 DEG
2663 Again26:      !
2665 Finish=0
2667 Offset=0
2669 !
```



```

2671 LINPUT "ASCII OR FLOATING POINT? (Enter A or F)",Input$
2673 IF UPC$(Input$)="A" THEN
2675     PRINT "Input ASCII (FORM4;) Data"
2677     GOTO Input_ascii
2679 ELSE
2681     PRINT "Input Floating Point (FORM3;) Data"
2683 END IF
2685 !
2687 Input_fp:      ! Input Floating Point
2689 IF Finish=1 THEN GOTO Exit_example26
2691 GOSUB Compute_trace
2693 !
2695 OUTPUT @Hwa;"FORM3; INPUDATA;"
2697 OUTPUT @Hwa_data2;Preamble,Size,Data(*)
2699 GOTO Input_fp
2701 !
2703 Input_ascii:   ! Input ASCII
2705 IF Finish=1 THEN GOTO Exit_example26
2707 GOSUB Compute_trace
2709 !
2711 OUTPUT @Hwa;"FORM4; INPUDATA;";
2713 OUTPUT @Hwa_data1;Data_ascii$(*)
2715 !
2717 GOTO Input_ascii
2719 !
2721 Finish:      !      Must Finish ASCII Trace Before Exit
2723 Finish=1
2725 RETURN
2727 !
2729 Exit_example26:  !
2731 LINPUT "Repeat Example? (Enter Y or N)",Input$
2733 IF UPC$(Input$)="Y" THEN GOTO Again26
2735 GOSUB Keys_off
2737 RETURN
2739 !
2741 Compute_trace:  !
2743 Offset=Offset-10
2745 FOR I=0 TO 200
2747     Data(I,0)=SIN(2*I+Offset)
2749     Data(I,1)=COS(2*I)
2751 NEXT I
2753 FOR I=0 TO 200
2755     Data_ascii$(I,0)=VAL$(Data(I,0))
2757     Data_ascii$(I,1)=VAL$(Data(I,1))
2759 NEXT I
2761 RETURN
2763 !
2765 Example27:      ! DELAY TABLE OPERATIONS *****
2767 PRINT
2769 PRINT "Example 27, Delay Table Operations"
2771 !
2773 DISP "Initializing System"
2775 OUTPUT @Hwa;"PRES; LIMP; SING; AUTO; DATI; DISPMATH; MARK1;"
2777 OUTPUT @Hwa;"FORM3; OUTPDATA;" ! current trace data is used for delay tbl.
2779 ENTER @Hwa_data2;Preamble,Size,Data(*) ! Get Data for Example
2781 !
2783 LINPUT "Press Return to Input Delay Table",Input$
2785 PRINT "Input Delay Table Data"
2787 OUTPUT @Hwa;"HOLD; FORM3; INPUDELA;"
2789 OUTPUT @Hwa_data2;Preamble,Size,Data(*)
2791 !

```

Programming Examples

```

2793 LINPUT "Press Return to Turn On Table Delay",Input$
2795 OUTPUT @Hwa;"TABD;"
2797 !
2799 LINPUT "Press Return to Turn Off Table Delay",Input$
2801 OUTPUT @Hwa;"COAD;" ! Or "WAVD;"
2803 !
2805 LINPUT "Press Return to Output Table Delay",Input$
2807 PRINT "Output Delay Table Data"
2809 OUTPUT @Hwa;"FORM3; OUTPDELA;"
2811 ENTER @Hwa_data2;Preamble,Size,Data(*)
2813 !
2815 LINPUT "Press Return to Store Delay Table to Disc",Input$
2817 PRINT "Store and Load Delay Table to Disc"
2819 OUTPUT @Hwa;"STOIINT; STOR; DELT; DISF""DELT"";"
2821 !
2823 LINPUT "Press Return to Load Delay Table form Disc",Input$
2825 OUTPUT @Hwa;"STOIINT; LOAD; DELT; DISF""DELT"";"
2827 RETURN
2829 !
2831 Example28: ! FASTCW Data Acquisition *****
2833 PRINT
2835 PRINT "Example 28, FASTCW Data Acquisition"
2837 !
2839 PRINT "Pulse Generator or External Trigger Source Required"
2841 LINPUT "Skip This Example? (Y or N)",Input$
2843 IF UPC$(Input$)<>"N" THEN GOTO Exit_example28
2845 LINPUT "Connect Pulse Gen or an external trigger source to HP 8510 EXT TRIGGER IN.",Input$
2847 !
2849 OUTPUT @Hwa;"PRES; CONT; SINP;"
2851 OUTPUT @Hwa;"CENT 10 GHz;" ! measurement frequency
2853 Again28: !
2855 GOSUB Blank_keys
2857 ON KEY 5 LABEL " NEXT EXAMPLE" GOSUB Exit_example28
2859 OUTPUT @Hwa;"FASC;"
2861 REPEAT ! WAIT UNTIL 8510 IS READY TO TAKE DATA.
2863 WAIT .001
2865 UNTIL BIT(SPOLL(@Hwa),2)
2869 !
2871 LINPUT "START PULSE GEN. OR EXTERNAL TRIGGER SOURCE THEN PRESS RETURN",Input$
2872 TRIGGER @Hwa ! ISSUE A SINGLE HPIB TRIGGER TO BEGIN FAST MODE.
2873 DISP "Collecting data, please wait..."
2875 !
2877 REDIM Form1_data(1:100,2) ! THE SIZE OF THIS ARRAY DETERMINES THE NUMBER
2879 ! OF POINTS MEASURED
2881 ENTER @Hwa_data2;Form1_data(*)! GET THE DATA, Continues when array is full
2883 !
2885 Data_collected: ! COLLECT DATA IN FORM 1 FORMAT.
2887 OUTPUT @Hwa;"SINP;" ! EXIT FROM FAST DATA MODE.
2889 OUTPUT @Hwa;"OUTPERRO;" ! CHECK ERROR STATUS.
2891 ENTER @Hwa_data1;Error_number,Input$
2893 PRINT "HP 8530A ERROR STATUS: ";Error_number,Input$
2895 PRINT SIZE(Form1_data,1);"Points Data Collected"
2897 !
2899 LINPUT "Press Return to Convert Data",Input$
2901 !

```

```

2903 ! This table is used to convert the exponent value from form1 data
2905 REAL Exp_tbl(0:255)
2907 Exp_tbl(0)=2^(-15) ! BUILD EXPONENT TABLE FOR DATA CONVERSION
2909 FOR I=0 TO 126
2911   Exp_tbl(I+1)=Exp_tbl(I)+Exp_tbl(I)
2913 NEXT I
2915 Exp_tbl(128)=2^(-143)
2917 FOR I=128 TO 254
2919   Exp_tbl(I+1)=Exp_tbl(I)+Exp_tbl(I)
2921 NEXT I
2923 !
2925 FOR N=1 TO SIZE(Form1_data,1) ! CONVERT THE DATA.
2927   Exponent=Exp_tbl(BINAND(Form1_data(N,2),255))
2929   Real=Form1_data(N,1)*Exponent
2931   Imag=Form1_data(N,0)*Exponent
2933   Lin_mag=20*LGT(SQRT(Real^2+Imag^2))
2935   IF N/20=INT(N/20) THEN PRINT "Point";N;" ";Lin_mag
2937 NEXT N
2939 !
2941 Exit_example28: !
2943 GOSUB Keys_off
2945 INPUT "Repeat Example? (Enter Y or N)",Input$
2947 IF UPC$(Input$)<>"Y" THEN
2949   RETURN
2951 ELSE
2953   GOTO Again28
2955 END IF
2957 !
2959 Example29: ! Test Port Power Flatness Cal *****
2961 PRINT
2963 PRINT "Example 29, Test Port Power Flatness Cal"
2965 PRINT "8360 Series Source and Power Meter On System Bus Required"
2967 INPUT "Skip this example? (Enter Y or N)",Input$
2969 IF UPC$(Input$)<>"N" THEN RETURN
2971 !
2973 DISP "Initializing System"
2975 OUTPUT @Hwa;"PRES; POWE 0; POIN51; STAR 2GHZ; STOP 20GHZ; ENTO;"
2977 PRINT "Frequency Range, Number of Points and Leveled Power Set"
2979 INPUT "Zero Power Meter and Connect to Port 1 (Press Return)",Input$
2981 !
2983 OUTPUT @Hwa;"CLES; CALF;" ! Bit 4 Set When Complete
2985 DISP "Flatness Calibration in Progress ..."
2987 REPEAT
2989   WAIT .5
2991   Ser_poll=SPOLL(@Hwa)
2993 UNTIL BIT(Ser_poll,4) ! flatness cal complete
2995 !
2997 OUTPUT @Hwa;"POWE -15; FLATON;" ! set test port power
2999 PRINT "Flatness On, Source 1 Power is now power at Port 1"
3001 !
3003 OUTPUT @Hwa;"USER1; STEP; SING; DATI; MAGO -15;"
3005 OUTPUT @Hwa;" DISPMATH; MARK1; SCAL 5; CONT;"
3007 OUTPUT @Hwa;"DWET .5; POWE;" ! dwell slows sweep for power meter
3009 PRINT "Change Source Power while Observing Trace AND Power Meter."
3011 LOCAL @Hwa
3013 RETURN
3015 !

```

Programming Examples

```
3017 Example30:      ! Receiver Power Cal *****
3019 PRINT
3021 PRINT "Example 30, Receiver Power Cal"
3023 PRINT "A Valid 2-20 Ghz Test Port Power Flatness Cal (Example 29)"
3025 PRINT "And 8510C Firmware >= Rev.C.07.00 Required."
3027 INPUT "Skip this example? (Enter Y or N)",Input$
3029 IF UPC$(Input$)<>"N" THEN RETURN
3031 DISP "Initializing System"
3033 OUTPUT @Hwa;"PRES; POIN101; STEP; STAR 2GHZ; STOP 20GHZ; MARK1; POWE -5; USER1;"
3035 INPUT "Connect Thru between Port 1 and Port 2 for Receiver Cal (Press Return)",Input$
3037 OUTPUT @Hwa;"CALRCVR; RCV0; RCVI; SAVR; CALS2;"
3039 OUTPUT @Hwa;"CHAN2; USER2; SPLI; MENUSTIM; SOFT1; SOFT8"
3041 LOCAL @Hwa
3043 PRINT "Calibrated DUT Input and Output Power now displayed (dbm)."
```

3045 INPUT "Turn Flatness On if leveled test port power is desired (Press Return)",Input\$

3047 OUTPUT @Hwa;"MARK1"

3049 LOCAL @Hwa

3051 INPUT "Set Marker to desired Frequency of Measurement for Power Domain (Press Return)",Input\$

3053 OUTPUT @Hwa;"POWD;"

3055 LOCAL @Hwa

3057 PRINT "Power Domain - Stimulus Start, Stop keys set Power"

3059 PRINT "Next Pt Higher / Lower keys select next Calibrated Frequency of Measurement"

3061 RETURN

3063 !

3065 ! *****

3067 ! End of Examples. The following subroutines are used by the examples

3069 ! *****

3071 !

3073 Run_mode: !

3075 INPUT "Run ALL Examples or a SINGLE Example? (Enter A or S)",Input\$

3077 IF UPC\$(Input\$)="A" THEN

3079 RETURN

3081 ELSE

3083 Choice: !

3085 INPUT "Enter the number of the example you wish to run. (1 to 30 or 0 to Quit)",Input\$

3087 IF Input\$="0" THEN

3089 LOCAL @Hwa

3091 STOP

3093 END IF

3095 IF Input\$="1" THEN GOSUB Example1

3097 IF Input\$="2" THEN GOSUB Example2

3099 IF Input\$="3" THEN GOSUB Example3

3101 IF Input\$="4" THEN GOSUB Example4

3103 IF Input\$="5" THEN GOSUB Example5

3105 IF Input\$="6" THEN GOSUB Example6

3107 IF Input\$="7" THEN GOSUB Example7

3109 IF Input\$="8" THEN GOSUB Example8

3111 IF Input\$="9" THEN GOSUB Example9

3113 IF Input\$="10" THEN GOSUB Example10

3115 IF Input\$="11" THEN GOSUB Example11

3117 IF Input\$="12" THEN GOSUB Example12

3119 IF Input\$="13" THEN GOSUB Example13

3121 IF Input\$="14" THEN GOSUB Example14

3123 IF Input\$="15" THEN GOSUB Example15

3125 IF Input\$="16" THEN GOSUB Example16

3127 IF Input\$="17" THEN GOSUB Example17

3129 IF Input\$="18" THEN GOSUB Example18

3131 IF Input\$="19" THEN GOSUB Example19

```

3133 IF Input$="20" THEN GOSUB Example20
3135 IF Input$="21" THEN GOSUB Example21
3137 IF Input$="22" THEN GOSUB Example22
3139 IF Input$="23" THEN GOSUB Example23
3141 IF Input$="24" THEN GOSUB Example24
3143 IF Input$="25" THEN GOSUB Example25
3145 IF Input$="26" THEN GOSUB Example26
3147 IF Input$="27" THEN GOSUB Example27
3149 IF Input$="28" THEN GOSUB Example28
3151 IF Input$="29" THEN GOSUB Example29
3153 IF Input$="30" THEN GOSUB Example30
3155 GOTO Choice
3157 END IF
3159 STOP
3161 !
3163 Wait_loop:WAIT .01
3165 GOTO Wait_loop
3167 ! *****
3169 Blank_keys: ! erases all the softkeys
3171 FOR I=1 TO 8
3173 ON KEY I LABEL "" GOSUB Do_nothing
3175 NEXT I
3177 RETURN
3179 ! *****
3181 Do_nothing: !
3183 WAIT .01
3185 RETURN
3187 ! *****
3189 Keys_off: !
3191 FOR I=1 TO 8
3193 OFF KEY I
3195 NEXT I
3197 RETURN
3199 ! *****
3201 No_analyzer: !
3203 DISP " *** 8510 Not Responding *** Check GPIB. Program Terminated."
3205 BEEP 200,.2
3207 END

```

